Let us consider a differential time-delay system defined by a vibrating string with an interior mass. See H. Mounier, J. Rudolph, M. Fliess, P. Rouchon, *Tracking Control of a Vibrating String with an Interior Mass viewed as Delay System*, ESAIM: Control, Optimisation and Calculus of Variations, 3 (1998), pp. 315-321.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Ore algebra *Alg*, where *Dt* is the differential operator w.r.t. t, $\sigma 1$ and $\sigma 2$ act as shift operators. Note that the parameters $\eta 1$, $\eta 2$, which are composed of the tensions, densities and the mass as defined in (Mounier et al., 1998), have to be declared in the definition of the Ore algebra:

```
> Alg := DefineOreAlgebra(diff=[Dt,t], dual_shift=[sigma1,y1],
> dual_shift=[sigma2,y2], polynom=[t,y1,y2], comm=[eta1,eta2],
> shift_action=[sigma1,t,tau1], shift_action=[sigma2,t,tau2]):
```

As in (Mounier et al., 1998), we study the case of position control on both boundaries and the case of a single control at one end only. Let us start with *two controls on the boundary.* We enter the system matrix R:

```
> R := evalm([[1,1,-1,-1,0,0],[Dt+eta1,Dt-eta1,-eta2,eta2,0,0],
> [sigma1^2,1,0,0,-sigma1,0],[0,0,1,sigma2^2,0,-sigma2]]);
```

$$R := \begin{bmatrix} 1 & 1 & -1 & -1 & 0 & 0 \\ Dt+\eta 1 & Dt-\eta 1 & -\eta 2 & \eta 2 & 0 & 0 \\ \sigma 1^2 & 1 & 0 & 0 & -\sigma 1 & 0 \\ 0 & 0 & 1 & \sigma 2^2 & 0 & -\sigma 2 \end{bmatrix}$$

Let us define the formal adjoint *R_adj* of R using an involution of *Alg*:

```
> R_adj := Involution(R, Alg):
```

We check controllability of the system by applying *Exti* to *R_adj*:

```
> st := time(): Ext1 := Exti(R_adj, Alg, 1): time()-st; Ext1[1];
```

$$1.340$$

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

We actually computed the first extension module ext^1 with values in *Alg* of the *Alg*-module N which is associated with *R_adj*. Since *Ext1*[1] is the identity matrix, we see that the *Alg*-module M which is associated with the system is torsion-free. This means that the vibrating string with interior mass is controllable and, equivalently, parametrizable. A parametrization of the system is given in *Ext1*[3]:

```
> Ext1[3];
```

$$\begin{bmatrix} 2\,\eta 2\,\sigma 2\,,\ -\sigma 2\,\sigma 1\,\eta 2\,,\ -\sigma 1\,Dt-\sigma 1\,\eta 2+\sigma 1\,\eta 1 \\ 0\,,\ \sigma 2\,\sigma 1\,\eta 2\,,\ \sigma 1\,\eta 2+\sigma 1\,Dt+\sigma 1\,\eta 1 \\ Dt\,\sigma 2+\eta 2\,\sigma 2+\sigma 2\,\eta 1\,,\ -\sigma 2\,\sigma 1\,\eta 1\,,\ 0 \\ -Dt\,\sigma 2+\eta 2\,\sigma 2-\sigma 2\,\eta 1\,,\ \sigma 2\,\sigma 1\,\eta 1\,,\ 2\,\sigma 1\,\eta 1 \\ 2\,\sigma 2\,\sigma 1\,\eta 2\,,\ -\sigma 2\,\sigma 1^2\,\eta 2+\eta 2\,\sigma 2\,,\ \eta 2-\eta 2\,\sigma 1^2-\sigma 1^2\,Dt+Dt+\sigma 1^2\,\eta 1+\eta 1 \\ -Dt\,\sigma 2^2+\eta 2\,\sigma 2^2-\eta 1\,\sigma 2^2+Dt+\eta 2+\eta 1\,,\ -\sigma 1\,\eta 1+\sigma 1\,\eta 1\,\sigma 2^2\,,\ 2\,\sigma 2\,\sigma 1\,\eta 1 \end{bmatrix}$$

So, the system can be parametrized by means of three arbitrary functions. We want to check now whether or not this parametrization is a *minimal* one. In order to do that, let us compute the rank of M:

```
> OreRank(R, Alg);
```

$$2$$

Hence, we know that there exist some parametrizations of the system which involve only two arbitrary functions. We find one minimal parametrization of the system as follows:

```
> P := MinimalParametrizations(R, Alg);
```

$$P := \left[\left[\begin{array}{cc} 2\,\eta 2\,\sigma 2 & -\sigma 2\,\sigma 1\,\eta 2 \\ 0 & \sigma 2\,\sigma 1\,\eta 2 \\ Dt\,\sigma 2 + \eta 2\,\sigma 2 + \sigma 2\,\eta 1 & -\sigma 2\,\sigma 1\,\eta 1 \\ -Dt\,\sigma 2 + \eta 2\,\sigma 2 - \sigma 2\,\eta 1 & \sigma 2\,\sigma 1\,\eta 1 \\ 2\,\sigma 2\,\sigma 1\,\eta 2 & -\sigma 2\,\sigma 1^2\,\eta 2 + \eta 2\,\sigma 2 \\ -Dt\,\sigma 2^2 + \eta 2\,\sigma 2^2 - \eta 1\,\sigma 2^2 + Dt + \eta 2 + \eta 1 & -\sigma 1\,\eta 1 + \sigma 1\,\eta 1\,\sigma 2^2 \end{array} \right], \right.$$

$$\left[\begin{array}{cc} 2\,\eta 2\,\sigma 2 & -\sigma 1\,Dt - \sigma 1\,\eta 2 + \sigma 1\,\eta 1 \\ 0 & \sigma 1\,\eta 2 + \sigma 1\,Dt + \sigma 1\,\eta 1 \\ Dt\,\sigma 2 + \eta 2\,\sigma 2 + \sigma 2\,\eta 1 & 0 \\ -Dt\,\sigma 2 + \eta 2\,\sigma 2 - \sigma 2\,\eta 1 & 2\,\sigma 1\,\eta 1 \\ 2\,\sigma 2\,\sigma 1\,\eta 2 & \eta 2 - \eta 2\,\sigma 1^2 - \sigma 1^2\,Dt + Dt + \sigma 1^2\,\eta 1 + \eta 1 \\ -Dt\,\sigma 2^2 + \eta 2\,\sigma 2^2 - \eta 1\,\sigma 2^2 + Dt + \eta 2 + \eta 1 & 2\,\sigma 2\,\sigma 1\,\eta 1 \end{array} \right],$$

$$\left. \left[\begin{array}{cc} -\sigma 2\,\sigma 1\,\eta 2 & -\sigma 1\,Dt - \sigma 1\,\eta 2 + \sigma 1\,\eta 1 \\ \sigma 2\,\sigma 1\,\eta 2 & \sigma 1\,\eta 2 + \sigma 1\,Dt + \sigma 1\,\eta 1 \\ -\sigma 2\,\sigma 1\,\eta 1 & 0 \\ \sigma 2\,\sigma 1\,\eta 1 & 2\,\sigma 1\,\eta 1 \\ -\sigma 2\,\sigma 1^2\,\eta 2 + \eta 2\,\sigma 2 & \eta 2 - \eta 2\,\sigma 1^2 - \sigma 1^2\,Dt + Dt + \sigma 1^2\,\eta 1 + \eta 1 \\ -\sigma 1\,\eta 1 + \sigma 1\,\eta 1\,\sigma 2^2 & 2\,\sigma 2\,\sigma 1\,\eta 1 \end{array} \right] \right]$$

Therefore, the first minimal parametrization of the system is defined by:

```
> evalm([[phi1(t)],[psi1(t)],[phi2(t)],[psi2(t)],[u(t)],[v(t)]])=
> ApplyMatrix(P[1], [xi1(t),x2(t)], Alg);
```

$$\left[\begin{array}{c} \phi 1(t) \\ \psi 1(t) \\ \phi 2(t) \\ \psi 2(t) \\ \mathrm{u}(t) \\ \mathrm{v}(t) \end{array} \right] =$$

$$[2\,\eta 2\,\xi 1(t-\tau 2) - \eta 2\,\mathrm{x2}(t-\tau 1-\tau 2)]$$
$$[\eta 2\,\mathrm{x2}(t-\tau 1-\tau 2)]$$
$$[\mathrm{D}(\xi 1)(t-\tau 2) + \eta 2\,\xi 1(t-\tau 2) + \eta 1\,\xi 1(t-\tau 2) - \eta 1\,\mathrm{x2}(t-\tau 1-\tau 2)]$$
$$[-\mathrm{D}(\xi 1)(t-\tau 2) + \eta 2\,\xi 1(t-\tau 2) - \eta 1\,\xi 1(t-\tau 2) + \eta 1\,\mathrm{x2}(t-\tau 1-\tau 2)]$$
$$[2\,\eta 2\,\xi 1(t-\tau 1-\tau 2) - \eta 2\,\mathrm{x2}(t-2\,\tau 1-\tau 2) + \eta 2\,\mathrm{x2}(t-\tau 2)]$$
$$[-\mathrm{D}(\xi 1)(t-2\,\tau 2) + \eta 2\,\xi 1(t-2\,\tau 2) - \eta 1\,\xi 1(t-2\,\tau 2) + \mathrm{D}(\xi 1)(t) + \eta 2\,\xi 1(t) + \eta 1\,\xi 1(t) - \eta 1\,\mathrm{x2}(t-\tau 1) + \eta 1\,\mathrm{x2}(t-\tau 1-2\,\tau 2)]$$

The second minimal parametrization of the system is defined by:

```
> evalm([[phi1(t)],[psi1(t)],[phi2(t)],[psi2(t)],[u(t)],[v(t)]])=
> ApplyMatrix(P[2], [xi1(t),x2(t)], Alg);
```

$$\begin{bmatrix} \phi 1(t) \\ \psi 1(t) \\ \phi 2(t) \\ \psi 2(t) \\ \mathrm{u}(t) \\ \mathrm{v}(t) \end{bmatrix} =$$

$$[2\,\eta 2\,\xi 1(t-\tau 2) - \mathrm{D}(x2)(t-\tau 1) - \eta 2\,\mathrm{x2}(t-\tau 1) + \eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[\eta 2\,\mathrm{x2}(t-\tau 1) + \mathrm{D}(x2)(t-\tau 1) + \eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[\mathrm{D}(\xi 1)(t-\tau 2) + \eta 2\,\xi 1(t-\tau 2) + \eta 1\,\xi 1(t-\tau 2)]$$
$$[-\mathrm{D}(\xi 1)(t-\tau 2) + \eta 2\,\xi 1(t-\tau 2) - \eta 1\,\xi 1(t-\tau 2) + 2\,\eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[2\,\eta 2\,\xi 1(t-\tau 1-\tau 2) + \eta 2\,\mathrm{x2}(t) - \eta 2\,\mathrm{x2}(t-2\,\tau 1) - \mathrm{D}(x2)(t-2\,\tau 1) + \mathrm{D}(x2)(t)$$
$$+ \eta 1\,\mathrm{x2}(t-2\,\tau 1) + \eta 1\,\mathrm{x2}(t)]$$
$$[-\mathrm{D}(\xi 1)(t-2\,\tau 2) + \eta 2\,\xi 1(t-2\,\tau 2) - \eta 1\,\xi 1(t-2\,\tau 2) + \mathrm{D}(\xi 1)(t) + \eta 2\,\xi 1(t)$$
$$+ \eta 1\,\xi 1(t) + 2\,\eta 1\,\mathrm{x2}(t-\tau 1-\tau 2)]$$

The third minimal parametrization of the system is defined by:

```
>  evalm([[phi1(t)],[psi1(t)],[phi2(t)],[psi2(t)],[u(t)],[v(t)]])=
>  ApplyMatrix(P[3], [xi1(t),x2(t)], Alg);
```

$$\begin{bmatrix} \phi 1(t) \\ \psi 1(t) \\ \phi 2(t) \\ \psi 2(t) \\ \mathrm{u}(t) \\ \mathrm{v}(t) \end{bmatrix} =$$

$$[-\eta 2\,\%1 - \mathrm{D}(x2)(t-\tau 1) - \eta 2\,\mathrm{x2}(t-\tau 1) + \eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[\eta 2\,\%1 + \eta 2\,\mathrm{x2}(t-\tau 1) + \mathrm{D}(x2)(t-\tau 1) + \eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[-\eta 1\,\%1]$$
$$[\eta 1\,\%1 + 2\,\eta 1\,\mathrm{x2}(t-\tau 1)]$$
$$[-\eta 2\,\xi 1(t-2\,\tau 1-\tau 2) + \eta 2\,\xi 1(t-\tau 2) + \eta 2\,\mathrm{x2}(t) - \eta 2\,\mathrm{x2}(t-2\,\tau 1)$$
$$- \mathrm{D}(x2)(t-2\,\tau 1) + \mathrm{D}(x2)(t) + \eta 1\,\mathrm{x2}(t-2\,\tau 1) + \eta 1\,\mathrm{x2}(t)]$$
$$[-\eta 1\,\xi 1(t-\tau 1) + \eta 1\,\xi 1(t-\tau 1-2\,\tau 2) + 2\,\eta 1\,\mathrm{x2}(t-\tau 1-\tau 2)]$$
$$\%1 := \xi 1(t-\tau 1-\tau 2)$$

Let us continue the study of the module properties of M. Let us check whether or not R has full row rank.

```
>  SyzygyModule(R, Alg);
```

$$\mathrm{INJ}(4)$$

We obtain that the rows of R are *Alg*-linearly independent, and thus, R has full row rank. Hence, we know that M is projective if and only if R admits a right-inverse.

```
>  RightInverse(R, Alg);
```

$$[]$$

Hence, M is not projective, which implies that M is not free, i.e., the vibrating string with interior mass is not a flat system. Another way to verify this is to compute the second and third extension modules ext^2 and ext^3 with values in *Alg* of the *Alg*-module N which is associated with *R_adj*:

```
> Exti(R_adj, Alg, 2);
```

$$\left[\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} \eta 2\,\sigma 2 & 0 & \sigma 1\,\eta 1 \\ \eta 2+\eta 1+Dt & -\sigma 1\,\eta 1 & 0 \\ 0 & \eta 2\,\sigma 2 & \eta 2+\eta 1+Dt \end{bmatrix}, \begin{bmatrix} -\sigma 1\,\eta 1 \\ -\eta 2 - Dt - \eta 1 \\ \eta 2\,\sigma 2 \end{bmatrix}\right]$$

```
> Exti(R_adj, Alg, 3);
```

$$\left[\begin{bmatrix} \sigma 2 \\ \sigma 1 \\ \eta 2+\eta 1+Dt \end{bmatrix}, \begin{bmatrix} 1 \end{bmatrix}, \mathrm{SURJ}(1)\right]$$

We see that the second extension module extˆ2 is zero, but the third extension module extˆ3 is different from zero. Therefore, M is a reflexive but not a projective Alg-module (we remember that M is projective if and only if extˆ i of R_adj is zero for $i = 1, 2, 3$). We can work out the obstructions to flatness. Let us find a polynomial π in the variable $\sigma 1$ such that the system is π-free (see H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995):

```
> PiPolynomial(R, Alg, [sigma1]);
```

$$[\sigma 1]$$

Let us find a polynomial π in the variable $\sigma 2$ such that the system is π-free:

```
> PiPolynomial(R, Alg, [sigma2]);
```

$$[\sigma 2]$$

Hence, if we invert $\sigma 1$ or $\sigma 2$, i.e., we allow ourselves to use time-advance operators, then by definition of the π-polynomial, the system becomes flat. A flat output for this system can be computed from a left-inverse of the minimal parametrization $P[1]$, where we allow $\sigma 1$ or $\sigma 2$ to appear in the denominators. Let us compute the annihilator of the cokernel of the minimal parametrization $P[1]$. We know from the theory that M is a torsion Alg-module.

```
> Ann1 := AnnExti(linalg[transpose](P[1]), Alg, 1);
```

$$Ann1 := [\sigma 2]$$

Let us compute a left-inverse of the minimal parametrization $P[1]$ by allowing $\sigma 2$ to appear in the denominators:

```
> L1 := LocalLeftInverse(P[1], Ann1, Alg);
```

$$L1 := \begin{bmatrix} 0 & 0 & \dfrac{1}{2\,\sigma 2\,\eta 2} & \dfrac{1}{2\,\sigma 2\,\eta 2} & 0 & 0 \\ 0 & \dfrac{\sigma 1}{\sigma 2\,\eta 2} & -\dfrac{\sigma 1}{\sigma 2\,\eta 2} & -\dfrac{\sigma 1}{\sigma 2\,\eta 2} & \dfrac{1}{\sigma 2\,\eta 2} & 0 \end{bmatrix}$$

We easily check that $L1$ is a left-inverse of $P[1]$:

```
> simplify(evalm(L1 &* P[1]));
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Thus, if we can invert $\sigma 2$, we obtain a flat output of the system defined by

$$(\xi_1 : \xi_2)^T = L1\,(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T,$$

which satisfies $(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T = P[1]\ (\xi_1 : \xi_2)^T$, where ϕ_1, ψ_1, ϕ_2, ψ_2, u, v are the system variables (Mounier et al., 1998). More precisely, we have:

```
> evalm([[xi1(t)],[xi2(t)]])=
> ApplyMatrix(L1, [phi1(t),psi1(t),phi2(t),psi2(t),u(t),v(t)], Alg);
```

$$\begin{bmatrix} \xi 1(t) \\ \xi 2(t) \end{bmatrix} = \begin{bmatrix} \frac{1}{2}\frac{\phi 2(t+\tau 2)}{\eta 2} + \frac{1}{2}\frac{\psi 2(t+\tau 2)}{\eta 2} \\ \frac{\psi 1(t+\tau 2-\tau 1)}{\eta 2} - \frac{\phi 2(t+\tau 2-\tau 1)}{\eta 2} - \frac{\psi 2(t+\tau 2-\tau 1)}{\eta 2} + \frac{\mathrm{u}(t+\tau 2)}{\eta 2} \end{bmatrix}$$

Let us point out that any multiplication of $(\xi_1 : \xi_2)^T$ by a matrix which is unimodular over $Alg[\sigma 2^{-1}]$ gives a new flat output of the system. For instance,

$$\theta 1 = 2\,\eta_2\,\sigma_2 \xi_1 = \phi_2 + \psi_2, \quad \theta 2 = \eta_2\,\sigma_2\,(\xi_2 + 2\,\sigma 1\,\xi_1) = \sigma 1\,\psi_1 = u$$

is a new flat output of the system. Let us prove this result. By substituting

$$(\xi_1 : \xi_2)^T = L1\,(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$$

into the parametrization $(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T = P[1]\ (\xi_1 : \xi_2)^T$, we obtain

$$(\phi_1 : \psi_2 : \phi_2 : \psi_2 : u : v)^T = Q1\,(\phi_1 : \psi_2 : \phi_2 : \psi_2 : u : v)^T,$$

where *Q1* is defined by:

```
> Q1 := simplify(evalm(P[1] &* L1));
```

$$Q1 :=$$

$$[0\,,\, -\sigma 1^2\,,\, 1+\sigma 1^2\,,\, 1+\sigma 1^2\,,\, -\sigma 1\,,\, 0]$$

$$[0\,,\, \sigma 1^2\,,\, -\sigma 1^2\,,\, -\sigma 1^2\,,\, \sigma 1\,,\, 0]$$

$$\left[0\,,\, -\frac{\sigma 1^2\,\eta 1}{\eta 2}\,,\, \frac{\eta 2+\eta 1+Dt+2\,\sigma 1^2\,\eta 1}{2\,\eta 2}\,,\, \frac{\eta 2+\eta 1+Dt+2\,\sigma 1^2\,\eta 1}{2\,\eta 2}\,,\, -\frac{\sigma 1\,\eta 1}{\eta 2}\,,\, 0\right]$$

$$\left[0\,,\, \frac{\sigma 1^2\,\eta 1}{\eta 2}\,,\, \frac{-Dt+\eta 2-\eta 1-2\,\sigma 1^2\,\eta 1}{2\,\eta 2}\,,\, \frac{-Dt+\eta 2-\eta 1-2\,\sigma 1^2\,\eta 1}{2\,\eta 2}\,,\, \frac{\sigma 1\,\eta 1}{\eta 2}\,,\, 0\right]$$

$$[0\,,\, -(-1+\sigma 1^2)\,\sigma 1\,,\, \sigma 1^3\,,\, \sigma 1^3\,,\, -\sigma 1^2+1\,,\, 0]$$

$$\left[0\,,\, \frac{\sigma 1^2\,\eta 1\,(-1+\sigma 2^2)}{\eta 2\,\sigma 2}\,,\right.$$
$$\frac{-Dt\,\sigma 2^2+\eta 2\,\sigma 2^2-\eta 1\,\sigma 2^2+Dt+\eta 2+\eta 1+2\,\sigma 1^2\,\eta 1-2\,\sigma 1^2\,\eta 1\,\sigma 2^2}{2\,\eta 2\,\sigma 2}\,,$$
$$\frac{-Dt\,\sigma 2^2+\eta 2\,\sigma 2^2-\eta 1\,\sigma 2^2+Dt+\eta 2+\eta 1+2\,\sigma 1^2\,\eta 1-2\,\sigma 1^2\,\eta 1\,\sigma 2^2}{2\,\eta 2\,\sigma 2}\,,$$
$$\left.\frac{\sigma 1\,\eta 1\,(-1+\sigma 2^2)}{\eta 2\,\sigma 2}\,,\, 0\right]$$

Let us point out that *Q1* is an idempotent of $Alg[\sigma 2^{-1}]^{6\times 6}$ (namely, $Q1$ satisfies $Q1^2 = Q1$), as we can easily check:

```
> simplify(evalm(Q1 &* Q1 - Q1));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Let us define the matrix *L1bis* as $(\theta 1 :\ \theta 2)^T = L1bis\ (\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$:

```
> L1bis := evalm([[0,0,1,1,0,0],[0,sigma1,0,0,1,0]]);
```

$$L1bis := \begin{bmatrix} 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & \sigma 1 & 0 & 0 & 1 & 0 \end{bmatrix}$$

```
> F1 := simplify(evalm(Factorize(evalm(sigma2*Q1), L1bis, Alg)/sigma2));
```

$$F1 :=$$
$$\left[1 + \sigma 1^2 ,\ -\sigma 1\right]$$
$$\left[-\sigma 1^2 ,\ \sigma 1\right]$$
$$\left[\frac{\eta 2 + \eta 1 + Dt + 2\,\sigma 1^2\,\eta 1}{2\,\eta 2} ,\ -\frac{\sigma 1\,\eta 1}{\eta 2}\right]$$
$$\left[\frac{-Dt + \eta 2 - \eta 1 - 2\,\sigma 1^2\,\eta 1}{2\,\eta 2} ,\ \frac{\sigma 1\,\eta 1}{\eta 2}\right]$$
$$\left[\sigma 1^3 ,\ -\sigma 1^2 + 1\right]$$
$$\left[\frac{-Dt\,\sigma 2^2 + \eta 2\,\sigma 2^2 - \eta 1\,\sigma 2^2 + Dt + \eta 2 + \eta 1 + 2\,\sigma 1^2\,\eta 1 - 2\,\sigma 1^2\,\eta 1\,\sigma 2^2}{2\,\eta 2\,\sigma 2} ,\right.$$
$$\left.\frac{\sigma 1\,\eta 1\,(-1 + \sigma 2^2)}{\eta 2\,\sigma 2}\right]$$

Therefore, the matrix *F1* is such that *Q1*= *F1 L1bis*, and thus, we have $(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T =$ Q1 $(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T = F1\ (L1bis\ (\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T) = F1\ (\theta 1 :\ \theta 2)^T$.

We let the reader check by himself that we obtain a parametrization of the system.

Moreover, we have $(\theta 1 : \theta 2)^T = L1bis\ (\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$ which proves that $(\theta 1 : \theta 2)^T$ is a flat output of the system. More precisely, we have:

```
> evalm([[theta1(t)],[theta2(t)]])=
> ApplyMatrix(L1bis, [phi1(t),psi1(t),phi2(t),psi2(t),u(t),v(t)], Alg);
```

$$\begin{bmatrix} \theta 1(t) \\ \theta 2(t) \end{bmatrix} = \begin{bmatrix} \phi 2(t) + \psi 2(t) \\ \psi 1(t - \tau 1) + \mathrm{u}(t) \end{bmatrix}$$

```
> evalm([[phi1(t)],[psi1(t)],[phi2(t)],[psi2(t)],[u(t)],[v(t)]])=
> ApplyMatrix(F1, [theta1(t),theta2(t)], Alg);
```

$$\begin{bmatrix} \phi1(t) \\ \psi1(t) \\ \phi2(t) \\ \psi2(t) \\ \mathrm{u}(t) \\ \mathrm{v}(t) \end{bmatrix} =$$

$$[\theta1(t) + \%1 - \theta2(t-\tau1)]$$
$$[-\%1 + \theta2(t-\tau1)]$$
$$\left[\frac{1}{2}\,\theta1(t) + \frac{1}{2}\,\frac{\eta1\,\theta1(t)}{\eta2} + \frac{1}{2}\,\frac{\mathrm{D}(\theta1)(t)}{\eta2} + \frac{\eta1\,\%1}{\eta2} - \frac{\eta1\,\theta2(t-\tau1)}{\eta2}\right]$$
$$\left[-\frac{1}{2}\,\frac{\mathrm{D}(\theta1)(t)}{\eta2} + \frac{1}{2}\,\theta1(t) - \frac{1}{2}\,\frac{\eta1\,\theta1(t)}{\eta2} - \frac{\eta1\,\%1}{\eta2} + \frac{\eta1\,\theta2(t-\tau1)}{\eta2}\right]$$
$$[\theta1(t-3\,\tau1) - \theta2(t-2\,\tau1) + \theta2(t)]$$
$$\Big[-\frac{1}{2}\,\frac{\mathrm{D}(\theta1)(t-\tau2)}{\eta2} + \frac{1}{2}\,\theta1(t-\tau2) - \frac{1}{2}\,\frac{\eta1\,\theta1(t-\tau2)}{\eta2} + \frac{1}{2}\,\frac{\mathrm{D}(\theta1)(t+\tau2)}{\eta2}$$
$$+ \frac{1}{2}\,\theta1(t+\tau2) + \frac{1}{2}\,\frac{\eta1\,\theta1(t+\tau2)}{\eta2} + \frac{\eta1\,\theta1(t+\tau2-2\,\tau1)}{\eta2} - \frac{\eta1\,\theta1(t-\tau2-2\,\tau1)}{\eta2}$$
$$- \frac{\eta1\,\theta2(t+\tau2-\tau1)}{\eta2} + \frac{\eta1\,\theta2(t-\tau2-\tau1)}{\eta2}\Big]$$
$$\%1 := \theta1(t-2\,\tau1)$$

We can repeat the same procedure for $P[2]$ and $P[3]$:

```
> Ann2 := AnnExti(linalg[transpose](P[2]), Alg, 1);
> Ann3 := AnnExti(linalg[transpose](P[3]), Alg, 1);
```

$$Ann2 := [\eta2 + \eta1 + Dt]$$
$$Ann3 := [\sigma1]$$

The annihilator of $P[3]$ contains only $\sigma1$. Let us compute a flat output by allowing the time-advance operator $\sigma1^{-1}$ to appear in the basis. Let us remark that this fact is not a problem for the main application of flatness, which is the motion planning problem. See (see H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995) for more details.

```
> L3 := LocalLeftInverse(P[3], Ann3, Alg);
```

$$L3 := \begin{bmatrix} 0 & 0 & 0 & \frac{\sigma2}{\sigma1\,\eta1} & 0 & -\frac{1}{\sigma1\,\eta1} \\ 0 & 0 & \frac{1}{2\,\sigma1\,\eta1} & \frac{1}{2\,\sigma1\,\eta1} & 0 & 0 \end{bmatrix}$$

The matrix $L3$ is a left-inverse of $P[3]$ over $Alg[\sigma1^{-1}]$, as we can check:

```
> simplify(evalm(L3 &* P[3]));
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Therefore, if we use the time-advance operator $\sigma1^{-1}$, we obtain that

$$(\xi_1 : \xi_2)^T = L3\,(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$$

is a flat output of the system which satisfies $(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T = P[3]\ (\xi_1 : \xi_2)^T$. More precisely, we have:

```
> evalm([[xi1(t)],[xi2(t)]])=
> ApplyMatrix(L3, [phi1(t),psi1(t),phi2(t),psi2(t),u(t),v(t)], Alg);
```

$$\begin{bmatrix} \xi1(t) \\ \xi2(t) \end{bmatrix} = \begin{bmatrix} \dfrac{\psi2(t+\tau1-\tau2)}{\eta1} - \dfrac{\mathrm{v}(t+\tau1)}{\eta1} \\ \dfrac{1}{2}\dfrac{\phi2(t+\tau1)}{\eta1} + \dfrac{1}{2}\dfrac{\psi2(t+\tau1)}{\eta1} \end{bmatrix}$$

Using trivial linear combinations of ξ_1 and ξ_2 with coefficients in $Alg[\sigma1^{-1}]$, we obtain that

$$(\zeta1 = \sigma2\,\psi_2 - v, \quad \zeta2 = \phi_2 + \psi_2)$$

is also a flat output of the system over $Alg[\sigma1^{-1}]$. Let us prove this result as we have done previously.

We first compute the matrix *Q3* defined by $Q3 = P[3] \circ L3$.

```
> Q3 := simplify(evalm(P[3] &* L3));
```

$$Q3 :=$$

$$\left[0,\,0,\,-\frac{Dt+\eta2-\eta1}{2\,\eta1},\,-\frac{2\,\eta2\,\sigma2^2+Dt+\eta2-\eta1}{2\,\eta1},\,0,\,\frac{\sigma2\,\eta2}{\eta1}\right]$$

$$\left[0,\,0,\,\frac{\eta2+\eta1+Dt}{2\,\eta1},\,\frac{2\,\eta2\,\sigma2^2+\eta2+\eta1+Dt}{2\,\eta1},\,0,\,-\frac{\sigma2\,\eta2}{\eta1}\right]$$

$$\left[0,\,0,\,0,\,-\sigma2^2,\,0,\,\sigma2\right]$$

$$\left[0,\,0,\,1,\,\sigma2^2+1,\,0,\,-\sigma2\right]$$

$$\left[0,\,0,\,-\frac{-\eta2+\eta2\,\sigma1^2+\sigma1^2\,Dt-Dt-\sigma1^2\,\eta1-\eta1}{2\,\sigma1\,\eta1},\right.$$

$$-\frac{-2\,\eta2\,\sigma2^2+2\,\eta2\,\sigma2^2\,\sigma1^2-\eta2+\eta2\,\sigma1^2+\sigma1^2\,Dt-Dt-\sigma1^2\,\eta1-\eta1}{2\,\sigma1\,\eta1},\,0,$$

$$\left.\frac{\eta2\,\sigma2\,(-1+\sigma1^2)}{\sigma1\,\eta1}\right]$$

$$\left[0,\,0,\,\sigma2,\,\sigma2^3,\,0,\,1-\sigma2^2\right]$$

Then, we define the matrix *L3bis* such that $(\zeta1 : \zeta2)^T = L3bis\ (\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$.

```
> L3bis := evalm([[0,0,0,sigma2,0,-1],[0,0,1,1,0,0]]);
```

$$L3bis := \begin{bmatrix} 0 & 0 & 0 & \sigma2 & 0 & -1 \\ 0 & 0 & 1 & 1 & 0 & 0 \end{bmatrix}$$

```
> F3 := simplify(evalm(Factorize(evalm(sigma1*Q3), L3bis, Alg)/sigma1));
```

$$F3 := \begin{bmatrix} -\dfrac{\sigma2\,\eta2}{\eta1} & -\dfrac{Dt+\eta2-\eta1}{2\,\eta1} \\ \dfrac{\sigma2\,\eta2}{\eta1} & \dfrac{\eta2+\eta1+Dt}{2\,\eta1} \\ -\sigma2 & 0 \\ \sigma2 & 1 \\ -\dfrac{\eta2\,\sigma2\,(-1+\sigma1^2)}{\sigma1\,\eta1} & -\dfrac{-\eta2+\eta2\,\sigma1^2+\sigma1^2\,Dt-Dt-\sigma1^2\,\eta1-\eta1}{2\,\sigma1\,\eta1} \\ -1+\sigma2^2 & \sigma2 \end{bmatrix}$$

Therefore, we have

$$\begin{aligned}(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T &= Q3\,(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T \\ &= F3\,(L3bis\,(\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T) = F3\,(\zeta 1 : \zeta 2)^T,\end{aligned}$$

where $(\zeta 1 : \zeta 2)^T = L3bis\ (\phi_1 : \psi_1 : \phi_2 : \psi_2 : u : v)^T$. Therefore, $(\zeta 1 : \zeta 2)^T$ is another flat output of the system. More precisely, we have:

```
> evalm([[zeta1(t)],[zeta2(t)]])=
> ApplyMatrix(L3bis, [phi1(t),psi1(t),phi2(t),psi2(t),u(t),v(t)], Alg);
```

$$\begin{bmatrix} \zeta 1(t) \\ \zeta 2(t) \end{bmatrix} = \begin{bmatrix} \psi 2(t-\tau 2) - \mathrm{v}(t) \\ \phi 2(t) + \psi 2(t) \end{bmatrix}$$

```
> evalm([[phi1(t)],[psi1(t)],[phi2(t)],[psi2(t)],[u(t)],[v(t)]])=
> ApplyMatrix(F3, [zeta1(t),zeta2(t)], Alg);
```

$$\begin{bmatrix} \phi 1(t) \\ \psi 1(t) \\ \phi 2(t) \\ \psi 2(t) \\ \mathrm{u}(t) \\ \mathrm{v}(t) \end{bmatrix} =$$

$$\left[-\frac{\eta 2\,\zeta 1(t-\tau 2)}{\eta 1} - \frac{1}{2}\,\frac{(\eta 2-\eta 1)\,\zeta 2(t)}{\eta 1} - \frac{1}{2}\,\frac{\mathrm{D}(\zeta 2)(t)}{\eta 1}\right]$$
$$\left[\frac{\eta 2\,\zeta 1(t-\tau 2)}{\eta 1} + \frac{1}{2}\,\frac{(\eta 2+\eta 1)\,\zeta 2(t)}{\eta 1} + \frac{1}{2}\,\frac{\mathrm{D}(\zeta 2)(t)}{\eta 1}\right]$$
$$[-\zeta 1(t-\tau 2)]$$
$$[\zeta 1(t-\tau 2) + \zeta 2(t)]$$
$$\left[\frac{\eta 2\,\zeta 1(t+\tau 1-\tau 2)}{\eta 1} - \frac{\eta 2\,\zeta 1(t-\tau 2-\tau 1)}{\eta 1} + \frac{1}{2}\,\frac{\eta 2\,\zeta 2(t+\tau 1)}{\eta 1} - \frac{1}{2}\,\frac{\eta 2\,\zeta 2(t-\tau 1)}{\eta 1}\right.$$
$$\left. - \frac{1}{2}\,\frac{\mathrm{D}(\zeta 2)(t-\tau 1)}{\eta 1} + \frac{1}{2}\,\frac{\mathrm{D}(\zeta 2)(t+\tau 1)}{\eta 1} + \frac{1}{2}\,\zeta 2(t-\tau 1) + \frac{1}{2}\,\zeta 2(t+\tau 1)\right]$$
$$[-\zeta 1(t) + \zeta 1(t-2\,\tau 2) + \zeta 2(t-\tau 2)]$$

Next we consider the system with a *single control at only one end of the spring*. The system matrix now is the following one. To obtain the system in (Mounier et al., 1998), one simply has to apply this matrix to the vector of system variables $(\phi_1 : \psi_1 : \psi_2 : u)^T$.

```
> R2 := evalm([[1,1,sigma2^2-1,0],[sigma1^2,1,0,-sigma1],
> [Dt+eta1,Dt-eta1,eta2*(1+sigma2^2),0]]);
```

$$R2 := \begin{bmatrix} 1 & 1 & \sigma 2^2 - 1 & 0 \\ \sigma 1^2 & 1 & 0 & -\sigma 1 \\ Dt + \eta 1 & Dt - \eta 1 & \eta 2\,(1+\sigma 2^2) & 0 \end{bmatrix}$$

We define a formal adjoint of *R2* using an involution of *Alg*:

```
> R2_adj := Involution(R2, Alg):
```

We use *Exti* again to check controllability of the system:

```
> st := time(): Ext1 := Exti(R2_adj, Alg, 1): time()-st; Ext1[1];
```

0.890

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

As in the preceding case, we see that the vibrating string with interior mass is controllable and, equivalently, parametrizable. A parametrization is defined by:

```
> map(collect, Ext1[3], {sigma1,Dt});
```

$$[((1 - \sigma 2^2)\, Dt + \eta 1\, \sigma 2^2 - \eta 1 + \eta 2 + \eta 2\, \sigma 2^2)\, \sigma 1]$$
$$[((\sigma 2^2 - 1)\, Dt - \eta 2 - \eta 1 - \eta 2\, \sigma 2^2 + \eta 1\, \sigma 2^2)\, \sigma 1]$$
$$[-2\, \sigma 1\, \eta 1]$$
$$\Big[((1 - \sigma 2^2)\, Dt + \eta 1\, \sigma 2^2 - \eta 1 + \eta 2 + \eta 2\, \sigma 2^2)\, \sigma 1^2 + (\sigma 2^2 - 1)\, Dt - \eta 2 - \eta 1 - \eta 2\, \sigma 2^2 + \eta 1\, \sigma 2^2\Big]$$

```
> evalm([[phi1(t)],[psi1(t)],[psi2(t)],[u(t)]])=Parametrization(R2, Alg);
```

$$\begin{bmatrix} \phi 1(t) \\ \psi 1(t) \\ \psi 2(t) \\ \mathrm{u}(t) \end{bmatrix} =$$
$$\Big[- \mathrm{D}(\xi_1)(t - \tau 1 - 2\,\tau 2) + \eta 1\, \%1 - \eta 1\, \xi_1(t - \tau 1) + \mathrm{D}(\xi_1)(t - \tau 1) + \eta 2\, \%1 + \eta 2\, \xi_1(t - \tau 1)\Big]$$
$$\Big[\mathrm{D}(\xi_1)(t - \tau 1 - 2\,\tau 2) - \mathrm{D}(\xi_1)(t - \tau 1) - \eta 2\, \%1 + \eta 1\, \%1 - \eta 1\, \xi_1(t - \tau 1) - \eta 2\, \xi_1(t - \tau 1)\Big]$$
$$[-2\, \eta 1\, \xi_1(t - \tau 1)]$$
$$\Big[- \eta 2\, \xi_1(t) - \eta 1\, \xi_1(t) - \mathrm{D}(\xi_1)(t) - \mathrm{D}(\xi_1)(t - 2\,\tau 2 - 2\,\tau 1) + \eta 1\, \xi_1(t - 2\,\tau 2 - 2\,\tau 1) + \mathrm{D}(\xi_1)(t - 2\,\tau 2) - \eta 1\, \xi_1(t - 2\,\tau 1) + \mathrm{D}(\xi_1)(t - 2\,\tau 1) + \eta 2\, \xi_1(t - 2\,\tau 2 - 2\,\tau 1) + \eta 1\, \xi_1(t - 2\,\tau 2) - \eta 2\, \xi_1(t - 2\,\tau 2) + \eta 2\, \xi_1(t - 2\,\tau 1)\Big]$$
$$\%1 := \xi_1(t - \tau 1 - 2\,\tau 2)$$

Of course, this parametrization is minimal. We compute the second extension module extˆ2 with values in *Alg* of the *Alg*-module N which is associated with *R2_adj*:

```
> Exti(R2_adj, Alg, 2);
```

$$[\begin{bmatrix} \sigma 1 \\ -\eta 2 - Dt + Dt\, \sigma 2^2 - \eta 2\, \sigma 2^2 + \eta 1\, \sigma 2^2 - \eta 1 \end{bmatrix}, \begin{bmatrix} 1 \end{bmatrix}, \mathrm{SURJ}(1)]$$

We see that the *Alg*-module M which is associated with the system is not projective, hence not free, and the vibrating string with interior mass is not a flat system. We can compute a π-polynomial in the variable $\sigma 1$ such that the system is π-free:

```
> PiPolynomial(R2, Alg, [sigma1]);
```

$$[\sigma 1]$$

When we allow the $\sigma 1$ to be inverted, i.e., if we introduce the time-advance operator $\sigma 1^{-1}$, the vibrating string with interior mass becomes a flat system. We compute a flat output of this system by constructing a left-inverse of the parametrization *Ext1*[3], where we allow the variable $\sigma 1$ in the denominators:

```
> L2 := LocalLeftInverse(Ext1[3], [sigma1], Alg);
```

$$L2 := \begin{bmatrix} 0 & 0 & -\frac{1}{2\,\sigma 1\,\eta 1} & 0 \end{bmatrix}$$

In fact, we have:

```
> evalm(L2 &* Ext1[3]);
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

Hence, remembering the variables ϕ_1, ψ_1, ψ_2, u of the system, we conclude that the vibrating string with interior mass is $\sigma 1$-free with basis ψ_2, i.e., ψ_2 is a flat output of the system, if we allow the time-advance operator $\sigma 1^{-1}$. Compare with (Mounier et al., 1998).

```
> Q := map(collect, evalm(Ext1[3] &* L2), Dt);
```

$$Q := \begin{bmatrix} 0, 0, -\frac{(-\sigma 2^2\,\sigma 1+\sigma 1)\,Dt}{2\,\sigma 1\,\eta 1} - \frac{\sigma 2^2\,\eta 2\,\sigma 1+\sigma 1\,\eta 1\,\sigma 2^2-\sigma 1\,\eta 1+\eta 2\,\sigma 1}{2\,\sigma 1\,\eta 1}, 0 \\ 0, 0, -\frac{(\sigma 2^2\,\sigma 1-\sigma 1)\,Dt}{2\,\sigma 1\,\eta 1} - \frac{-\sigma 1\,\eta 1-\eta 2\,\sigma 1-\sigma 2^2\,\eta 2\,\sigma 1+\sigma 1\,\eta 1\,\sigma 2^2}{2\,\sigma 1\,\eta 1}, 0 \\ 0, 0, 1, 0 \\ 0, 0, -\frac{(-1-\sigma 2^2\,\sigma 1^2+\sigma 2^2+\sigma 1^2)\,Dt}{2\,\sigma 1\,\eta 1} - \frac{-\eta 2-\eta 1-\sigma 1^2\,\eta 1+\eta 2\,\sigma 1^2+\sigma 1^2\,\eta 1\,\sigma 2^2+\eta 1\,\sigma 2^2-\eta 2\,\sigma 2^2+\sigma 2^2\,\eta 2\,\sigma 1^2}{2\,\sigma 1\,\eta 1}, 0 \end{bmatrix}$$

Finally, the third column of the matrix Q gives the parametrization of the system variables ϕ_1, ψ_1, ψ_2, u in function of the flat output ψ_2. In particular, we have the following parametrization:

```
> phi1(t)=ApplyMatrix(Q, [0,0,psi2(t),0], Alg)[1,1];
```

$$\phi 1(t) = \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t-2\,\tau 2)}{\eta 1} - \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t)}{\eta 1} - \frac{1}{2}\,\frac{\eta 2\,\psi 2(t-2\,\tau 2)}{\eta 1} - \frac{1}{2}\,\psi 2(t-2\,\tau 2) + \frac{1}{2}\,\psi 2(t) - \frac{1}{2}\,\frac{\eta 2\,\psi 2(t)}{\eta 1}$$

```
> psi1(t)=ApplyMatrix(Q, [0,0,psi2(t),0], Alg)[2,1];
```

$$\psi 1(t) = -\frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t-2\,\tau 2)}{\eta 1} + \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t)}{\eta 1} + \frac{1}{2}\,\psi 2(t) + \frac{1}{2}\,\frac{\eta 2\,\psi 2(t)}{\eta 1} + \frac{1}{2}\,\frac{\eta 2\,\psi 2(t-2\,\tau 2)}{\eta 1} - \frac{1}{2}\,\psi 2(t-2\,\tau 2)$$

```
> psi2(t)=ApplyMatrix(Q, [0,0,psi2(t),0], Alg)[3,1];
```

$$\psi 2(t) = \psi 2(t)$$

```
> u(t)=ApplyMatrix(Q, [0,0,psi2(t),0], Alg)[4,1];
```

$$
\begin{aligned}
\mathrm{u}(t) = {} & \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t+\tau 1)}{\eta 1} + \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t-\tau 1-2\,\tau 2)}{\eta 1} - \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t+\tau 1-2\,\tau 2)}{\eta 1} \\
& - \frac{1}{2}\,\frac{\mathrm{D}(\psi 2)(t-\tau 1)}{\eta 1} + \frac{1}{2}\,\frac{\eta 2\,\psi 2(t+\tau 1)}{\eta 1} + \frac{1}{2}\,\psi 2(t+\tau 1) + \frac{1}{2}\,\psi 2(t-\tau 1) \\
& - \frac{1}{2}\,\frac{\eta 2\,\psi 2(t-\tau 1)}{\eta 1} - \frac{1}{2}\,\psi 2(t-\tau 1-2\,\tau 2) - \frac{1}{2}\,\psi 2(t+\tau 1-2\,\tau 2) \\
& + \frac{1}{2}\,\frac{\eta 2\,\psi 2(t+\tau 1-2\,\tau 2)}{\eta 1} - \frac{1}{2}\,\frac{\eta 2\,\psi 2(t-\tau 1-2\,\tau 2)}{\eta 1}
\end{aligned}
$$